
USER DOCUMENTATION GALAHAD Optimization Library version 5.4

1 SUMMARY

Given real n by n symmetric matrices H and S (with S diagonally dominant), a real n vector c and scalars f and ∆ > 0,

this package uses an extended Krylov-subspace method to find a global minimizer of the quadratic objective

function 1
2 xT Hx+ cT x+ f , where the vector x is required to satisfy the constraint ‖x‖S ≤ ∆, and where the S-

norm of x is ‖x‖S =
√

xT Sx. This problem commonly occurs as a trust-region subproblem in nonlinear optimization

calculations. The matrix S need not be provided in the commonly-occurring ℓ2-trust-region case for which S = I, the

n by n identity matrix.

A factorization of the matrix H (and S, if it is required) will be used, so this package is most suited for the case where

such a factorization may be found efficiently. If this is not the case, the package GALAHAD GLTR may be preferred.

ATTRIBUTES — Versions: GALAHAD TREK single, GALAHAD TREK double. Uses: GALAHAD CLOCK, GALAH-

AD SYMBOLS, GALAHAD SPACE, GALAHAD NORMS, GALAHAD ROOTS, GALAHAD SPECFILE, GALAHAD SMT, GALAHAD RAND,

GALAHAD TRS, GALAHAD SLS, GALAHAD MOP Date: November 2025. Origin: H. Al Daas and N. I. M. Gould, Ruther-

ford Appleton Laboratory. Language: Fortran 95 + TR 15581 or Fortran 2003. Parallelism: Some options may use

OpenMP and its runtime library.

2 HOW TO USE THE PACKAGE

The package is available with single, double and (if available) quadruple precision reals, and either 32-bit or 64-bit

integers. Access to the 32-bit integer, single precision version requires the USE statement

USE GALAHAD TREK single

with the obvious substitution GALAHAD TREK double, GALAHAD TREK quadruple, GALAHAD TREK single 64, GALA-

HAD TREK double 64 and GALAHAD TREK quadruple 64 for the other variants.

If it is required to use more than one of the modules at the same time, the derived types SMT TYPE, TREK control type,

TREK inform type, TREK data type, (Section 2.4) and the subroutines TREK initialize, TREK solve, TREK -

terminate (Section 2.5) and TREK read specfile (Section 2.7) must be renamed on one of the USE statements.

2.1 Matrix storage formats

The matrices H and (if required) S may be stored in a variety of input formats.

2.1.1 Dense storage format

The matrix H is stored as a compact dense matrix by rows, that is, the values of the entries of each row in turn are

stored in order within an appropriate real one-dimensional array. Since H is symmetric, only the lower triangular part

(that is the part hi j for 1 ≤ j ≤ i ≤ n) need be held. In this case the lower triangle should be stored by rows, that is

component i∗ (i−1)/2+ j of the storage array H%val will hold the value hi j (and, by symmetry, h ji) for 1 ≤ j ≤ i ≤ n.

The same is true for S if it is used.

All use is subject to the conditions of a BSD-3-Clause License.

2.1.2 Sparse co-ordinate storage format

Only the nonzero entries of the matrices are stored. For the l-th entry of H, 1 ≤ j ≤ i ≤ n, its row index i, column index

j and value hi j are stored in the l-th components of the integer arrays H%row, H%col and real array H%val, respectively.

Note that only the entries in the lower triangle should be stored. The same scheme may be used for S if it is required.

2.1.3 Sparse row-wise storage format

Again only the nonzero entries are stored, but this time they are ordered so that those in row i appear directly before

those in row i+ 1. For the i-th row of H, the i-th component of the integer array H%ptr holds the position of the first

entry in this row, while H%ptr (m+ 1) holds the total number of entries plus one. The column indices j, 1 ≤ j ≤ i,

and values hi j of the entries in the i-th row are stored in components l = H%ptr(i), . . . ,H%ptr (i+1)−1 of the integer

array H%col, and real array H%val, respectively. Note that as before only the entries in the lower triangle should be

stored. For sparse matrices, this scheme almost always requires less storage than its predecessor. This scheme may

also be used for S if it is required.

2.1.4 Diagonal storage format

If H is diagonal (i.e., hi j = 0 for all 1 ≤ i 6= j ≤ n) only the diagonals entries hii, 1 ≤ i ≤ n, need be stored, and the first

n components of the array H%val may be used for the purpose. The same applies to S if it is required.

2.1.5 Scaled-identity-matrix storage format

If H is a scalar multiple of the identity matrix (i.e., hii = h11 and hi j = 0 for all 1 ≤ i 6= j ≤ n) only the first diagonal

entry h11 needs be stored, and the first component of the array H%val may be used for the purpose. The same applies

to S if it is required.

2.1.6 Identity-matrix storage format

If H is the identity matrix (i.e., hii = 1 and hi j = 0 for all 1 ≤ i 6= j ≤ n), no explicit entries needs be stored. The same

would be true for S, but S need not be provided if S = I.

2.1.7 Zero-matrix storage format

If H = 0 (i.e., hi j = 0 for all 1 ≤ i, j ≤ n), no explicit entries needs be stored. This is not relevant for S, as S is required

to be non-singular.

2.2 Real and integer kinds

We use the terms integer and real to refer to the fortran keywords REAL(rp ) and INTEGER(ip ), where rp and

ip are the relevant kind values for the real and integer types employed by the particular module in use. The former

are equivalent to default REAL for the single precision versions, DOUBLE PRECISION for the double precision cases

and quadruple-precision if 128-bit reals are available, and correspond to rp = real32, rp = real64 and rp =

real128 respectively as defined by the fortran iso fortran env module. The latter are default (32-bit) and long

(64-bit) integers, and correspond to ip = int32 and ip = int64, respectively, again from the iso fortran env

module.

All use is subject to the conditions of a BSD-3-Clause License.
See http://galahad.rl.ac.uk/galahad-www/cou.html for full details.

2.3 Parallel usage

OpenMP may be used by the GALAHAD TREK package to provide parallelism for some solvers in shared memory

environments. See the documentation for the GALAHAD package SLS for more details. To run in parallel, OpenMP

must be enabled at compilation time by using the correct compiler flag (usually some variant of -openmp). The number

of threads may be controlled at runtime by setting the environment variable OMP NUM THREADS.

MPI may also be used by the package to provide parallelism for some solvers in a distributed memory environment.

To use this form of parallelism, MPI must be enabled at runtime by using the correct compiler flag (usually some

variant of -lmpi). Although the MPI process will be started automatically when required, it should be stopped by the

calling program once no further use of this form of parallelism is needed. Typically, this will be via statements of the

form

CALL MPI_INITIALIZED( flag, ierr )

IF ( flag ) CALL MPI_FINALIZE( ierr )

The code may be compiled and run in serial mode.

2.4 The derived data types

Six derived data types are accessible from the package.

2.4.1 The derived data type for holding matrices

The derived data type SMT TYPE is used to hold the matrices H and perhaps S. The components of SMT TYPE used here

are:

m is a scalar component of type INTEGER(ip ), that holds the number of rows in the matrix.

n is a scalar component of type INTEGER(ip ), that holds the number of columns in the matrix.

ne is a scalar variable of type INTEGER(ip ), that holds the number of matrix entries.

type is a rank-one allocatable array of type default CHARACTER, that is used to indicate the matrix storage scheme

used. Its precise length and content depends on the type of matrix to be stored.

val is a rank-one allocatable array of type REAL(rp ) and dimension at least ne, that holds the values of the entries.

Each pair of off-diagonal entries hi j = h ji of the symmetric matrix H is represented as a single entry (see §2.1.1–

2.1.3). Any duplicated entries that appear in the sparse co-ordinate or row-wise schemes will be summed.

row is a rank-one allocatable array of type INTEGER(ip ), and dimension at least ne, that may hold the row indices

of the entries. (see §2.1.2).

col is a rank-one allocatable array of type INTEGER(ip ), and dimension at least ne, that may hold the column

indices of the entries (see §2.1.2–2.1.3).

ptr is a rank-one allocatable array of type INTEGER(ip ), and dimension at least n + 1, that may holds the pointers

to the first entry in each row (see §2.1.3).

2.4.2 The derived data type for holding control parameters

The derived data type TREK control type is used to hold controlling data. Default values may be obtained by calling

TREK initialize (see Section 2.5.1). The components of TREK control type are:

All use is subject to the conditions of a BSD-3-Clause License.
See http://galahad.rl.ac.uk/galahad-www/cou.html for full details.

error is a scalar variable of type INTEGER(ip ), that holds the stream number for error messages. Printing of error

messages in TREK solve and TREK terminate is suppressed if error≤ 0. The default is error = 6.

out is a scalar variable of type INTEGER(ip ), that holds the stream number for informational messages. Printing

of informational messages in TREK solve is suppressed if out< 0. The default is out = 6.

print level is a scalar variable of type INTEGER(ip ), that is used to control the amount of informational output

which is required. No informational output will occur if print level≤ 0. If print level = 1 a single line

of output will be produced for each iteration of the process. If print level ≥ 2 this output will be increased

to provide significant detail of each iteration. The default is print level = 0.

eks max is a scalar variable of type INTEGER(ip ), that specifies the maximum dimension of the extended Krylov

space employed. If a negative value is given, this will be replaced by 100. The default is eks max = - 1.

it max is a scalar variable of type INTEGER(ip ), that is used to specify the maximum number of iterations allowed.

If a negative value is given, this will be replaced by 100. The default is it max = - 1.

f is a scalar variable of type default REAL(rp ), that gives the value of the constant term f in the quadratic

objective function. This value has no effect on the computed minimizer x. The default is f = 0.0.

reduction is a scalar variable of type REAL(rp ), that holds the value of the reduction factor for a suggested sub-

sequent trust-region radius (see control%next radius). The suggested radius will be reduction times the

smaller of the current radius and ‖x‖S at the output x. The default is reduction = 0.5.

stop residual is a scalar variable of type REAL(rp ), that holds the value of the stopping tolerance used by the

algorithm. The iteration stops as soon as x and λ are found to satisfy ‖(H+ λS)x+ c‖ < stop residual

×max(1,‖c‖). The default is stop residual = 10
√

u, where u is EPSILON(1.0 rp )

reorthogonalize is a scalar variable of type default LOGICAL, that may be set .TRUE. if the user wishes the pack-

age to reorthogonalise the generated basis of the extended Krylov space at every iteration. This can be very

expensive, and is generally not warranted. The default is reorthogonalize = .FALSE. which is our recom-

mendation.

s version 52 is a scalar variable of type default LOGICAL, that may be set .TRUE. if the user wishes to use Algorithm

5.2 in the paper to generate the extended Krylov space recurrences when a non-unit S is given, and .FALSE. if

those from Algorithm B.3 should be used instead. In practice, there is very little difference in performance and

accuracy. The default is s version 52 = .TRUE..

perturb c is a scalar variable of type default LOGICAL, that may be set .TRUE. if the user wishes the package make a

tiny pseudo-random perturbations to the components of the term c to try to protect from the so-called (probability

zero) “hard case”. Perturbations are generally not needed, and should only be used in very exceptional cases.

The default is perturb c = .FALSE..

stop check all orders is a scalar variable of type default LOGICAL, that may be set .TRUE. if the user wishes the

package to check for termination for each new member of the extended Krylov space. Such checks incur some

extra cost, and experience shows that testing every second member is sufficient, using stop check all orders

= .FALSE.. The default is stop check all orders = .FALSE..

new radius is a scalar variable of type default LOGICAL, that if .TRUE. will resolve the problem with the previous H,

c and (if provided) S, but with the smaller radius, radius, provided to TREK solve. The default is new radius

= .FALSE..

new values is a scalar variable of type default LOGICAL, that if PRESENT and .TRUE. will resolve the problem with

the data structures set for the previous H and (if provided) S, but with new values H%val, C and/or S%val

provided to TREK solve. The default is new values = .FALSE..

All use is subject to the conditions of a BSD-3-Clause License.

space critical is a scalar variable of type default LOGICAL, that may be set .TRUE. if the user wishes the package

to allocate as little internal storage as possible, and .FALSE. otherwise. The package may be more efficient if

space critical is set .FALSE.. The default is space critical = .FALSE..

deallocate error fatal is a scalar variable of type default LOGICAL, that may be set .TRUE. if the user wishes the

package to return to the user in the unlikely event that an internal array deallocation fails, and .FALSE. if the

package should be allowed to try to continue. The default is deallocate error fatal = .FALSE..

linear solver is a scalar variable of type default CHARACTER and length 30, that specifies the external package to

be used to solve any symmetric positive-definite linear system involving H that might arise. Current possible

choices are ’sils’, ’ma27’, ’ma57’, ’ma77’, ’ma86’, ’ma87’, ’ma97’, ssids, ’pardiso’, ’wsmp’, ’sytr’,

’potr’ and ’pbtr’ although only ’sytr’, ’potr’, ’pbtr’ and, for OMP 4.0-compliant compilers, ’ssids’

are installed by default. See the documentation for the GALAHAD package SLS for further details. The default

is linear solver = ’ssids’, but we recommend ’pbtr’ instead if H is banded with a small bandwidth.

linear solver for S is a scalar variable of type default CHARACTER and length 30, that specifies the external pack-

age to be used to solve any symmetric positive-definite linear system involving the optional S that might arise.

Current possible choices are ’sils’, ’ma27’, ’ma57’, ’ma77’, ’ma86’, ’ma87’, ’ma97’, ssids, ’pardiso’,

’wsmp’, ’sytr’, ’potr’ and ’pbtr’ although only ’sytr’, ’potr’, ’pbtr’ and, for OMP 4.0-compliant

compilers, ’ssids’ are installed by default. See the documentation for the GALAHAD package SLS for further

details. The default is linear solver = ’ssids’, but we recommend ’pbtr’ instead if S is banded with a

small bandwidth.

prefix is a scalar variable of type default CHARACTER and length 30, that may be used to provide a user-selected

character string to preface every line of printed output. Specifically, each line of output will be prefaced by the

string prefix(2:LEN(TRIM(prefix))-1), thus ignoring the first and last non-null components of the supplied

string. If the user does not want to preface lines by such a string, they may use the default prefix = "".

SLS control is a scalar variable of type SLS control type that is used to control various aspects of the factorization

package SLS. See the documentation for GALAHAD SLS for more details.

SLS S control is a scalar variable of type SLS control type that is used to control various aspects of the factoriza-

tion package SLS when applied to S. See the documentation for GALAHAD SLS for more details.

TRS control is a scalar variable of type TRS control type that is used to control various aspects of the diagonal

trust-region solver from the package TRS. See the documentation for GALAHAD TRS for more details.

2.4.3 The derived data type for holding timing information

The derived data type TREK time type is used to hold elapsed CPU and system clock times for the various parts of

the calculation. The components of TREK time type are:

total is a scalar variable of type REAL(rp ), that gives the total CPU time spent in the package.

assemble is a scalar variable of type REAL(rp ), that gives the CPU time spent assembling the matrices H and (if

provided) S from their constituent parts.

analyse is a scalar variable of type REAL(rp ), that gives the CPU time spent analysing the matrices H and (if

provided) S prior to factorization.

factorize is a scalar variable of type REAL(rp ), that gives the CPU time spent factorizing the matrices H and (if

provided) S.

solve is a scalar variable of type REAL(rp ), that gives the CPU time spent using the factors to solve relevant linear

equations.

All use is subject to the conditions of a BSD-3-Clause License.

clock total is a scalar variable of type REAL(rp ), that gives the total elapsed system clock time spent in the

package.

clock assemble is a scalar variable of type REAL(rp ), that gives the elapsed system clock time spent assembling

the matrices H and (if provided) S from their constituent parts.

clock analyse is a scalar variable of type REAL(rp ), that gives the elapsed system clock time spent analysing the

matrices H and (if provided) S prior to factorization.

clock factorize is a scalar variable of type REAL(rp ), that gives the elapsed system clock time spent factorizing

the required matrices.

clock solve is a scalar variable of type REAL(rp ), that gives the elapsed system clock time spent using the factors

to solve relevant linear equations.

2.4.4 The derived data type for holding informational parameters

The derived data type TREK inform type is used to hold parameters that give information about the progress and

needs of the algorithm. The components of TREK inform type are:

status is a scalar variable of type INTEGER(ip ), that gives the current status of the algorithm. See Section 2.6 for

details.

alloc status is a scalar variable of type INTEGER(ip ), that gives the status of the last internal array allocation or

deallocation. This will be 0 if status = 0.

bad alloc is a scalar variable of type default CHARACTER and length 80, that gives the name of the last internal array

for which there were allocation or deallocation errors. This will be the null string if status = 0.

iter is a scalar variable of type INTEGER(ip ), that gives the total number of iterations required.

n vec is a scalar variable of type INTEGER(ip ), that gives the number of orthogonal vectors required (the dimension

of the extended-Krylov subspace).

obj is a scalar variable of type REAL(rp ), that holds the value of the objective function 1
2 xT Hx+ cTx.

x norm is a scalar variable of type REAL(rp ), that holds the value of ‖x‖S.

multiplier is a scalar variable of type REAL(rp ), that holds the value of the Lagrange multiplier λ associated with

the constraint.

radius is a scalar variable of type REAL(rp ), that holds the value of the current radius.

next radius is a scalar variable of type REAL(rp ), that holds the value of the proposed next radius to be used if the

current radius proves to be too large (see inform%reduction).

error is a scalar variable of type REAL(rp ), that holds the value of the norm of the maximum relative residual error,

‖(H+λS)x+ c‖/max(1,‖c‖).

time is a scalar variable of type TREK time type whose components are used to hold elapsed CPU and system clock

times for the various parts of the calculation (see Section 2.4.3).

SLS inform is a scalar variable of type SLS inform type, that holds informational parameters concerning the anal-

ysis, factorization and solution phases for H performed by the GALAHAD sparse matrix factorization package

SLS. See the documentation for the package SLS for details of the derived type SLS inform type.

All use is subject to the conditions of a BSD-3-Clause License.

SLS S inform is a scalar variable of type SLS inform type, that holds informational parameters concerning the anal-

ysis, factorization and solution phases for S, if present, performed by the GALAHAD sparse matrix factorization

package SLS. See the documentation for the package SLS for details of the derived type SLS inform type.

TRS inform is a scalar variable of type TRS inform type, that holds informational parameters concerning the diag-

onal trust-region subroutine contained in the GALAHAD refinement package TRS. See the documentation for the

package TRS for details of the derived type TRS inform type.

2.4.5 The derived data type for holding problem data

The derived data type TREK data type is used to hold all the data for a particular problem between calls of TREK

procedures. This data should be preserved, untouched, from the initial call to TREK initialize to the final call to

TREK terminate.

2.5 Argument lists and calling sequences

There are three procedures for user calls (see Section 2.7 for further features):

1. The subroutine TREK initialize is used to set default values and initialize private data.

2. The subroutine TREK solve is called to solve the problem.

3. The subroutine TREK terminate is provided to allow the user to automatically deallocate array components of

the private data, allocated by TREK solve, at the end of the solution process.

We use square brackets [ ] to indicate OPTIONAL arguments.

2.5.1 The initialization subroutine

Default values are provided as follows:

CALL TREK initialize( data, control, inform )

data is a scalar INTENT(INOUT)argument of type TREK data type (see Section 2.4.5). It is used to hold data about

the problem being solved.

control is a scalar INTENT(OUT)argument of type TREK control type (see Section 2.4.2). On exit, control con-

tains default values for the components as described in Section 2.4.2. These values should only be changed after

calling TREK initialize.

inform is a scalar INTENT(OUT) argument of type TREK inform type (see Section 2.4.4). A successful call to

TREK initialize is indicated when the component status has the value 0. For other return values of status,

see Section 2.6.

2.5.2 The optimization problem solution subroutine

The trust-region solution algorithm is called as follows:

CALL TREK solve( n, H, C, radius, X, data, control, inform[, S])

n is a scalar INTENT(IN) argument of type INTEGER(ip ), that must be set to the number of unknowns, n. Re-

striction: n > 0.

All use is subject to the conditions of a BSD-3-Clause License.

H is scalar INTENT(IN) argument of type SMT TYPE that holds the Hessian matrix H. The following components

are used here:

H%type is an allocatable array of rank one and type default CHARACTER, that is used to indicate the storage

scheme used. If the dense storage scheme (see Section 2.1.1) is used, the first five components of H%type

must contain the string DENSE. For the sparse co-ordinate scheme (see Section 2.1.2), the first ten com-

ponents of H%type must contain the string COORDINATE, for the sparse row-wise storage scheme (see

Section 2.1.3), the first fourteen components of H%type must contain the string SPARSE BY ROWS, and for

the diagonal storage scheme (see Section 2.1.4), the first eight components of H%type must contain the

string DIAGONAL, for the scaled-identity matrix storage scheme (see Section 2.1.5), the first fifteen com-

ponents of H%type must contain the string SCALED IDENTITY, for the identity matrix storage scheme (see

Section 2.1.6), the first eight components of H%type must contain the string IDENTITY, and for the zero

matrix storage scheme (see Section 2.1.7), the first four components of H%type must contain the string

ZERO.

For convenience, the procedure SMT put may be used to allocate sufficient space and insert the required

keyword into H%type. For example, if we wish to store S using the co-ordinate scheme, we may simply

CALL SMT_put( H%type, ’COORDINATE’ )

See the documentation for the GALAHAD package SMT for further details on the use of SMT put.

H%ne is a scalar variable of type INTEGER(ip ), that holds the number of entries in the lower triangular part

of H in the sparse co-ordinate storage scheme (see Section 2.1.2). It need not be set for any of the other

three schemes.

H%val is a rank-one allocatable array of type REAL(rp ), that holds the values of the entries of the lower

triangular part of the Hessian matrix H in any of the storage schemes discussed in Section 2.1.

H%row is a rank-one allocatable array of type INTEGER(ip ), that holds the row indices of the lower triangular

part of H in the sparse co-ordinate storage scheme (see Section 2.1.2). It need not be allocated for any of

the other three schemes.

H%col is a rank-one allocatable array variable of type INTEGER(ip ), that holds the column indices of the

lower triangular part of H in either the sparse co-ordinate (see Section 2.1.2), or the sparse row-wise (see

Section 2.1.3) storage scheme. It need not be allocated when the dense or diagonal storage schemes are

used.

H%ptr is a rank-one allocatable array of dimension n+1 and type INTEGER(ip ), that holds the starting position

of each row of the lower triangular part of H, as well as the total number of entries plus one, in the sparse

row-wise storage scheme (see Section 2.1.3). It need not be allocated when the other schemes are used.

C is an array INTENT(IN) argument of dimension n and type REAL(rp ), whose i-th entry holds the component ci

of the vector c for the objective function.

radius is a scalar INTENT(IN) variable of type default REAL(rp ), that must be set on initial entry to the value of

the radius of the trust-region constraint, ∆. Restriction: ∆ > 0.

X is an array INTENT(OUT) argument of dimension n and type REAL(rp ), that holds an estimate of the solution x

of the problem on exit.

data is a scalar INTENT(INOUT)argument of type TREK data type (see Section 2.4.5). It is used to hold data about

the problem being solved. It must not have been altered by the user since the last call to TREK initialize.

control is a scalar INTENT(IN) argument of type TREK control type. (see Section 2.4.2). Default values may be

assigned by calling TREK initialize prior to the first call to TREK solve.

All use is subject to the conditions of a BSD-3-Clause License.

inform is a scalar INTENT(INOUT)argument of type TREK inform type (see Section 2.4.4) whose components need

not be set on entry. A successful call to TREK solve is indicated when the component status has the value 0.

For other return values of status, see Section 2.6.

S is an OPTIONAL scalar INTENT(IN) argument of type SMT TYPE that holds the diagonally dominant scaling

matrix S. It need only be set if S 6= I and in this case the following components are used:

S%type is an allocatable array of rank one and type default CHARACTER, that is used to indicate the storage

scheme used. If the dense storage scheme (see Section 2.1.1) is used, the first five components of S%type

must contain the string DENSE. For the sparse co-ordinate scheme (see Section 2.1.2), the first ten com-

ponents of S%type must contain the string COORDINATE, for the sparse row-wise storage scheme (see

Section 2.1.3), the first fourteen components of S%type must contain the string SPARSE BY ROWS, and for

the diagonal storage scheme (see Section 2.1.4), the first eight components of S%type must contain the

string DIAGONAL, for the scaled-identity matrix storage scheme (see Section 2.1.5), the first fifteen com-

ponents of S%type must contain the string SCALED IDENTITY, for the identity matrix storage scheme (see

Section 2.1.6), the first eight components of S%type must contain the string IDENTITY, and for the zero

matrix storage scheme (see Section 2.1.7), the first four components of S%type must contain the string

ZERO.

For convenience, the procedure SMT put may be used to allocate sufficient space and insert the required

keyword into S%type. For example, if we wish to store S using the co-ordinate scheme, we may simply

CALL SMT_put( M%type, ’COORDINATE’ )

See the documentation for the GALAHAD package SMT for further details on the use of SMT put.

S%ne is a scalar variable of type INTEGER(ip ), that holds the number of entries in the lower triangular part

of S in the sparse co-ordinate storage scheme (see Section 2.1.2). It need not be set for any of the other

three schemes.

S%val is a rank-one allocatable array of type REAL(rp ), that holds the values of the entries of the lower

triangular part of the scaling matrix S in any of the storage schemes discussed in Section 2.1.

S%row is a rank-one allocatable array of type INTEGER(ip ), that holds the row indices of the lower triangular

part of S in the sparse co-ordinate storage scheme (see Section 2.1.2). It need not be allocated for any of

the other three schemes.

S%col is a rank-one allocatable array variable of type INTEGER(ip ), that holds the column indices of the

lower triangular part of S in either the sparse co-ordinate (see Section 2.1.2), or the sparse row-wise (see

Section 2.1.3) storage scheme. It need not be allocated when the dense or diagonal storage schemes are

used.

S%ptr is a rank-one allocatable array of dimension n+1 and type INTEGER(ip ), that holds the starting position

of each row of the lower triangular part of S, as well as the total number of entries plus one, in the sparse

row-wise storage scheme (see Section 2.1.3). It need not be allocated when the other schemes are used.

If S is absent, the ℓ2-norm, ‖x‖2 =
√

xT x, will be employed.

2.5.3 The termination subroutine

All previously allocated arrays are deallocated as follows:

CALL TREK terminate( data, control, inform )

data is a scalar INTENT(INOUT)argument of type TREK data type exactly as for TREK solve that must not have

been altered by the user since the last call to TREK initialize. On exit, array components will have been

deallocated.

All use is subject to the conditions of a BSD-3-Clause License.

control is a scalar INTENT(IN)argument of type TREK control type exactly as for TREK solve.

inform is a scalar INTENT(OUT)argument of type TREK inform type exactly as for TREK solve. Only the compo-

nent status will be set on exit, and a successful call to TREK terminate is indicated when this component

status has the value 0. For other return values of status, see Section 2.6.

2.6 Warning and error messages

A negative value of inform%status on exit from TREK solve or TREK terminate indicates that an error might have

occurred. No further calls should be made until the error has been corrected. Possible values are:

-1. An allocation error occurred. A message indicating the offending array is written on unit control%error, and the

returned allocation status and a string containing the name of the offending array are held in inform%alloc -

status and inform%bad alloc, respectively.

-2. A deallocation error occurred. A message indicating the offending array is written on unit control%error and

the returned allocation status and a string containing the name of the offending array are held in inform%alloc -

status and inform%bad alloc, respectively.

-3. (TREK solve only) One of the restrictions n > 0 radius > 0, H%n 6= n, S%n 6= n or the radius has not decreased

when resolve = /true/ has been violated.

-9. (TREK solve only) The analysis phase of the factorization of one of the matrices H and (if provided) S failed.

-10. (TREK solve only) The factorization of one of the matrices H and (if provided) S failed.

-11. (TREK solve only) The solve phase involving one of the matrices H and (if provided) S failed.

-15. (TREK solve only) The matrix S appears not to be diagonally dominant.

-16. (TREK solve only) The problem is so ill-conditioned that further progress is impossible.

-18. (TREK solve only) Too many iterations have been required. This may happen if control%eks max is too small,

but may also be symptomatic of a badly scaled problem.

-31. (TREK solve only) A resolve call has been made before an initial call (see control%new radius and control%new values).

-38. (TREK solve only) An error occurred in a call to an LAPACK subroutine.

2.7 Further features

In this section, we describe an alternative means of setting control parameters, that is components of the variable

control of type TREK control type (see Section 2.4.2), by reading an appropriate data specification file using the

subroutine TREK read specfile. This facility is useful as it allows a user to change TREK control parameters without

editing and recompiling programs that call TREK.

A specification file, or specfile, is a data file containing a number of ”specification commands”. Each command

occurs on a separate line, and comprises a ”keyword”, which is a string (in a close-to-natural language) used to identify

a control parameter, and an (optional) ”value”, which defines the value to be assigned to the given control parameter.

All keywords and values are case insensitive, keywords may be preceded by one or more blanks but values must not

contain blanks, and each value must be separated from its keyword by at least one blank. Values must not contain more

than 30 characters, and each line of the specfile is limited to 80 characters, including the blanks separating keyword

and value.

The portion of the specification file used by TREK read specfile must start with a ”BEGIN TREK” command and

end with an ”END” command. The syntax of the specfile is thus defined as follows:

All use is subject to the conditions of a BSD-3-Clause License.
See http://galahad.rl.ac.uk/galahad-www/cou.html for full details.

( .. lines ignored by TREK_read_specfile .. )

BEGIN TREK

keyword value

....... .....

keyword value

END

( .. lines ignored by TREK_read_specfile .. )

where keyword and value are two strings separated by (at least) one blank. The “BEGIN TREK” and “END” delimiter

command lines may contain additional (trailing) strings so long as such strings are separated by one or more blanks,

so that lines such as

BEGIN TREK SPECIFICATION

and

END TREK SPECIFICATION

are acceptable. Furthermore, between the “BEGIN TREK” and “END” delimiters, specification commands may occur in

any order. Blank lines and lines whose first non-blank character is ! or * are ignored. The content of a line after a !

or * character is also ignored (as is the ! or * character itself). This provides an easy manner to ”comment out” some

specification commands, or to comment specific values of certain control parameters.

The value of a control parameters may be of three different types, namely integer, logical or real. Integer and real

values may be expressed in any relevant Fortran integer and floating-point formats (respectively). Permitted values for

logical parameters are ”ON”, ”TRUE”, ”.TRUE.”, ”T”, ”YES”, ”Y”, or ”OFF”, ”NO”, ”N”, ”FALSE”, ”.FALSE.” and ”F”.

Empty values are also allowed for logical control parameters, and are interpreted as ”TRUE”.

The specification file must be open for input when TREK read specfile is called, and the associated device

number passed to the routine in device (see below). Note that the corresponding file is REWINDed, which makes it

possible to combine the specifications for more than one program/routine. For the same reason, the file is not closed

by TREK read specfile.

Control parameters corresponding to the components TRS control, SLS control and IR control may be changed

by including additional sections enclosed by “BEGIN TRS” and “END TRS”, and “BEGIN SLS” and “END SLS”, respec-

tively. See the specification sheets for the packages GALAHAD TRS and GALAHAD SLS

for further details.

2.7.1 To read control parameters from a specification file

Control parameters may be read from a file as follows:

CALL TREK_read_specfile( control, device )

control is a scalar INTENT(INOUT)argument of type TREK control type (see Section 2.4.2). Default values should

have already been set, perhaps by calling TREK initialize. On exit, individual components of control may

have been changed according to the commands found in the specfile. Specfile commands and the component

(see Section 2.4.2) of control that each affects are given in Table 2.1.

device is a scalar INTENT(IN)argument of type INTEGER(ip ), that must be set to the unit number on which the

specfile has been opened. If device is not open, control will not be altered and execution will continue, but

an error message will be printed on unit control%error.

All use is subject to the conditions of a BSD-3-Clause License.
See http://galahad.rl.ac.uk/galahad-www/cou.html for full details.

command component of control value type

error-printout-device %error integer

printout-device %out integer

print-level %print level integer

maximum-subspace-dimension %eks max integer

maximum-number-of-iterations %it max integer

constant-term-in-objective %f real

radius-reduction-factor %reduction real

residual-accuracy %stop residual real

reorthogonalize-vectors %reorthogonalize logical

s-version-52 %s version 52 logical

perturb-c %perturb-c logical

stop-check-all-orders %stop-check-all-orders logical

space-critical %space critical logical

deallocate-error-fatal %deallocate error fatal logical

linear-equation-solver %linear solver character

linear-equation-solver-for-S %linear solver for S character

output-line-prefix %prefix character

Table 2.1: Specfile commands and associated components of control.

2.8 Information printed

If control%print level is positive, information about the progress of the algorithm will be printed on unit control-

%out. If control%print level = 1, a single line of output will be produced for each iteration of the process.

This will include the size of the subspace, the norm of the current estimate of x and the radius, the current es-

timate of the shift λ, the error ‖(H+ λS)x+ c‖/max(1,‖c‖), and the current value of the objective function. If

control%print level ≥ 2, this output will be increased to provide significant detail of each iteration. This extra

output includes times for various phases.

3 GENERAL INFORMATION

Use of common: None.

Workspace: Provided automatically by the module.

Other routines called directly: None.

Other modules used directly: TREK solve calls the GALAHAD packages GALAHAD CLOCK, GALAHAD SYMBOLS, GA-

LAHAD SPACE, GALAHAD NORMS, GALAHAD ROOTS, GALAHAD SPECFILE, GALAHAD SMT, GALAHAD RAND, GALAH-

AD TRS, GALAHAD SLS, and GALAHAD MOP.

Input/output: Output is under control of the arguments control%error, control%out and control%print level.

Restrictions: n > 0, ∆ > 0.

Portability: ISO Fortran 95 + TR 15581 or Fortran 2003. The package is thread-safe.

All use is subject to the conditions of a BSD-3-Clause License.
See http://galahad.rl.ac.uk/galahad-www/cou.html for full details.

4 METHOD

The required solution x∗ necessarily satisfies the optimality condition Hx∗+λ∗Sx∗+c= 0, where λ∗ ≥ 0 is a Lagrange

multiplier corresponding to the constraint ‖x‖S ≤ ∆. In addition in all cases, the matrix H+λ∗S will be positive semi-

definite; in most instances it will actually be positive definite, but in special “hard” cases singularity is a possibility.

The method is iterative, and is based upon building a solution approximation from an orthogonal basis of the evolving

extended Krylov subspaces K2m+1(H,c) = span{c,H−1c,Hc,H−2c,H2c, . . . ,H−mc,Hmc} as m increases. The key

observations are (i) the manifold of solutions to the optimality system

(H+λI)x(λ) =−c

as a function of σ is of approximately very low rank, (ii) the subspace K2m+1(H,c) rapidly gives a very good ap-

proximation to this manifold, (iii) it is straightforward to build an orthogonal basis of K2m+1(H,c) using short-term

recurrences and a single factorization of H, and (iv) solutions to the trust-region subproblem restricted to elements

of the orthogonal subspace may be found very efficiently using effective high-order root-finding methods. The fact

that the second element in the subspace is H−1c means that it is easy to check for the interior-solution possibility

x =−H−1c that occurs when such a x satisfies ‖x‖≤ ∆. Coping with general scalings S is a straightforward extension

so long as factorization of S is also possible.

Reference: The method is described in detail in

H. Al Daas and N. I. M. Gould. Extended-Krylov-subspace methods for trust-region and norm-regularization sub-

problems. Preprint STFC-P-2025-002, Rutherford Appleton Laboratory, Oxfordshire, England.

5 EXAMPLE OF USE

Suppose we wish to solve a problem in 10,000 unknowns, whose data is

H =





−2 1

1 −2 .
. . .

. −2 1

1 −2



.



and f = 0,

starting with a radius ∆ = 10, but then gradually halving the radius at every subsequent stage. Then we may use the

following code:

! THIS VERSION: GALAHAD 5.4 - 2025-11-15 AT 10:00 GMT.

PROGRAM GALAHAD_TREK_EXAMPLE

! double precision version

USE GALAHAD_TREK_double, ONLY: SMT_type, SMT_put, TREK_control_type, &

TREK_inform_type, TREK_data_type, TREK_initialize, TREK_solve, &

TREK_terminate

USE GALAHAD_NORMS_double, ONLY: TWO_NORM

IMPLICIT NONE

INTEGER, PARAMETER :: wp = KIND( 1.0D+0 ) ! set precision

INTEGER, PARAMETER :: n = 10000, m = 4, h_ne = 2 * n - 1

TYPE ( SMT_type ) :: H

REAL ( KIND = wp ), DIMENSION( n ) :: C

REAL ( KIND = wp ) :: radius

REAL ( KIND = wp ), DIMENSION( n ) :: X

TYPE ( TREK_control_type ) :: control

TYPE ( TREK_inform_type ) :: inform

All use is subject to the conditions of a BSD-3-Clause License.

TYPE ( TREK_data_type ) :: data

INTEGER :: stat, i

H%n = n ; H%ne = h_ne

CALL SMT_put( H%type, ’COORDINATE’, stat ) ! Specify co-ordinate for H

ALLOCATE( H%val( H%ne ), H%row( H%ne ), H%col( H%ne ) )

DO i = 1, n

H%row( i ) = i ; H%col( i ) = i ; H%val( i ) = - 2.0_wp

END DO

DO i = 1, n - 1

H%row( n + i ) = i + 1 ; H%col( n + i ) = i ; H%val( n + i ) = 1.0_wp

END DO

C = 1.0_wp ! c is a vector of ones

radius = 10.0_wp ! initial radius

CALL TREK_initialize( data, control, inform )

control%linear_solver = ’pbtr ’

DO i = 1, m ! loop over a sequence of decreasing radii

control%new_radius = i > 1

inform%time%clock_total = 0.0 ! reset time

CALL TREK_solve( n, H, C, radius, X, data, control, inform )

IF ( inform%status == 0 ) THEN

WRITE( 6, "( 1X, I0, ’ vectors required, error = ’, ES11.4 )" ) &

inform%n_vec, inform%error

WRITE( 6, "( ’ radius, ||x||, f, multiplier =’, 2ES11.4, 2ES12.4 )" ) &

radius, TWO_NORM( X ), inform%obj, inform%multiplier

IF ( i < m ) radius = inform%next_radius ! pick the next (smaller) radius

ELSE

WRITE( 6, "( ’ error exit, status = ’, I0 )" ) inform%status

STOP

END IF

WRITE( 6, "( ’ total time TREK = ’, F0.2 )" ) inform%time%clock_total

END DO

CALL TREK_terminate( data, control, inform )

DEALLOCATE( H%type, H%row, H%col, H%val )

END PROGRAM GALAHAD_TREK_EXAMPLE

This produces the following output:

12 vectors required, error = 4.5524E-08

radius, ||x||, f, multiplier = 1.0000E+01 1.0000E+01 -1.0000E+03 1.0000E+01

total time TREK = .01

12 vectors required, error = 3.7551E-10

radius, ||x||, f, multiplier = 5.0000E+00 5.0000E+00 -5.0000E+02 2.0000E+01

total time TREK = .00

12 vectors required, error = 4.4371E-12

radius, ||x||, f, multiplier = 2.5000E+00 2.5000E+00 -2.5000E+02 4.0000E+01

total time TREK = .00

12 vectors required, error = 6.0771E-14

radius, ||x||, f, multiplier = 1.2500E+00 1.2500E+00 -1.2500E+02 8.0000E+01

total time TREK = .00

All use is subject to the conditions of a BSD-3-Clause License.

